In[5]:=
```
NotebookFileName[]
```
Out[5]= `C:\Dropbox\Work\myweb\Courses\Math_pages\Math_304\Gram_Schmidt_Monomials.nb`

Legendre polynomials $P_0(x)$, ..., $P_9(x)$

Below is a table of the inner products of the first ten monomials:

In[∘]:=
```
TableForm[Table[Integrate[x^j x^k, {x, -1, 1}], {j, 0, 9}, {k, 0, 9}],
 TableHeadings → {{"1", "x", "x^2", "x^3", "x^4", "x^5", "x^6", "x^7", "x^8", "x^9"},
   {"1", "x", "x^2", "x^3", "x^4", "x^5", "x^6", "x^7", "x^8", "x^9"}}]
```
Out[∘]//TableForm=

	1	x	x^2	x^3	x^4	x^5	x^6	x^7	x^8	x^9
1	2	0	$\frac{2}{3}$	0	$\frac{2}{5}$	0	$\frac{2}{7}$	0	$\frac{2}{9}$	0
x	0	$\frac{2}{3}$	0	$\frac{2}{5}$	0	$\frac{2}{7}$	0	$\frac{2}{9}$	0	$\frac{2}{11}$
x^2	$\frac{2}{3}$	0	$\frac{2}{5}$	0	$\frac{2}{7}$	0	$\frac{2}{9}$	0	$\frac{2}{11}$	0
x^3	0	$\frac{2}{5}$	0	$\frac{2}{7}$	0	$\frac{2}{9}$	0	$\frac{2}{11}$	0	$\frac{2}{13}$
x^4	$\frac{2}{5}$	0	$\frac{2}{7}$	0	$\frac{2}{9}$	0	$\frac{2}{11}$	0	$\frac{2}{13}$	0
x^5	0	$\frac{2}{7}$	0	$\frac{2}{9}$	0	$\frac{2}{11}$	0	$\frac{2}{13}$	0	$\frac{2}{15}$
x^6	$\frac{2}{7}$	0	$\frac{2}{9}$	0	$\frac{2}{11}$	0	$\frac{2}{13}$	0	$\frac{2}{15}$	0
x^7	0	$\frac{2}{9}$	0	$\frac{2}{11}$	0	$\frac{2}{13}$	0	$\frac{2}{15}$	0	$\frac{2}{17}$
x^8	$\frac{2}{9}$	0	$\frac{2}{11}$	0	$\frac{2}{13}$	0	$\frac{2}{15}$	0	$\frac{2}{17}$	0
x^9	0	$\frac{2}{11}$	0	$\frac{2}{13}$	0	$\frac{2}{15}$	0	$\frac{2}{17}$	0	$\frac{2}{19}$

Mathematica has its own command for to orthonormalize functions.

The command

In[∘]:=
```
x^Range[0, 9]
```
Out[∘]= $\{1, x, x^2, x^3, x^4, x^5, x^6, x^7, x^8, x^9\}$

lists the first ten monomials. The following command will give us the orthonormalized monomials. The logic of this command is that we give the formula for the inner product in the special form:

In[∘]:=
```
Integrate[#1 * #2, {x, -1, 1}] &[f[x], g[x]]
```
Out[∘]= $\int_{-1}^{1} f[x] \times g[x] \, dx$

Below I name the list of orthonormalized polynomials **OrNoPo9**

In[1]:=
```
OrNoPo9 =
  Expand[FullSimplify[Orthogonalize[x^Range[0, 9], Integrate[#1 * #2, {x, -1, 1}] &]]]
```

Out[1]=

$$\Big\{\frac{1}{\sqrt{2}},\ \sqrt{\frac{3}{2}}\,x,\ -\frac{\sqrt{\frac{5}{2}}}{2}+\frac{3}{2}\sqrt{\frac{5}{2}}\,x^2,\ -\frac{3}{2}\sqrt{\frac{7}{2}}\,x+\frac{5}{2}\sqrt{\frac{7}{2}}\,x^3,\ \frac{9}{8\sqrt{2}}-\frac{45\,x^2}{4\sqrt{2}}+\frac{105\,x^4}{8\sqrt{2}},$$

$$\frac{15}{8}\sqrt{\frac{11}{2}}\,x-\frac{35}{4}\sqrt{\frac{11}{2}}\,x^3+\frac{63}{8}\sqrt{\frac{11}{2}}\,x^5,\ -\frac{5\sqrt{\frac{13}{2}}}{16}+\frac{105}{16}\sqrt{\frac{13}{2}}\,x^2-\frac{315}{16}\sqrt{\frac{13}{2}}\,x^4+\frac{231}{16}\sqrt{\frac{13}{2}}\,x^6,$$

$$-\frac{35}{16}\sqrt{\frac{15}{2}}\,x+\frac{315}{16}\sqrt{\frac{15}{2}}\,x^3-\frac{693}{16}\sqrt{\frac{15}{2}}\,x^5+\frac{429}{16}\sqrt{\frac{15}{2}}\,x^7,$$

$$\frac{35\sqrt{\frac{17}{2}}}{128}-\frac{315}{32}\sqrt{\frac{17}{2}}\,x^2+\frac{3465}{64}\sqrt{\frac{17}{2}}\,x^4-\frac{3003}{32}\sqrt{\frac{17}{2}}\,x^6+\frac{6435}{128}\sqrt{\frac{17}{2}}\,x^8,$$

$$\frac{315}{128}\sqrt{\frac{19}{2}}\,x-\frac{1155}{32}\sqrt{\frac{19}{2}}\,x^3+\frac{9009}{64}\sqrt{\frac{19}{2}}\,x^5-\frac{6435}{32}\sqrt{\frac{19}{2}}\,x^7+\frac{12\,155}{128}\sqrt{\frac{19}{2}}\,x^9\Big\}$$

The orthonormalized polynomials have norms 1. To prove this calculate:

In[∘]:=
```
TableForm[Table[Integrate[OrNoPo9⟦j⟧ OrNoPo9⟦k⟧, {x, -1, 1}], {j, 1, 10}, {k, 1, 10}]]
```

Out[∘]//TableForm=

1	0	0	0	0	0	0	0	0	0
0	1	0	0	0	0	0	0	0	0
0	0	1	0	0	0	0	0	0	0
0	0	0	1	0	0	0	0	0	0
0	0	0	0	1	0	0	0	0	0
0	0	0	0	0	1	0	0	0	0
0	0	0	0	0	0	1	0	0	0
0	0	0	0	0	0	0	1	0	0
0	0	0	0	0	0	0	0	1	0
0	0	0	0	0	0	0	0	0	1

It is common to introduce the polynomials that have value 1 at x=1. Now calculate the values at x=1 for all polynomials in **OrNoPo9**

In[∘]:=
```
OrNoPo9 /. {x → 1}
```

Out[∘]=

$$\Big\{\frac{1}{\sqrt{2}},\ \sqrt{\frac{3}{2}},\ \sqrt{\frac{5}{2}},\ \sqrt{\frac{7}{2}},\ \frac{3}{\sqrt{2}},\ \sqrt{\frac{11}{2}},\ \sqrt{\frac{13}{2}},\ \sqrt{\frac{15}{2}},\ \sqrt{\frac{17}{2}},\ \sqrt{\frac{19}{2}}\Big\}$$

In[2]:=
```
LePo9 = Expand[1/(OrNoPo9 /. {x → 1}) OrNoPo9]
```

Out[2]=

$$\left\{1, x, -\frac{1}{2}+\frac{3x^2}{2}, -\frac{3x}{2}+\frac{5x^3}{2}, \frac{3}{8}-\frac{15x^2}{4}+\frac{35x^4}{8}, \frac{15x}{8}-\frac{35x^3}{4}+\frac{63x^5}{8}, -\frac{5}{16}+\frac{105x^2}{16}-\frac{315x^4}{16}+\frac{231x^6}{16}, -\frac{35x}{16}+\frac{315x^3}{16}-\frac{693x^5}{16}+\frac{429x^7}{16}, \frac{35}{128}-\frac{315x^2}{32}+\frac{3465x^4}{64}-\frac{3003x^6}{32}+\frac{6435x^8}{128}, \frac{315x}{128}-\frac{1155x^3}{32}+\frac{9009x^5}{64}-\frac{6435x^7}{32}+\frac{12155x^9}{128}\right\}$$

Next we calculate that the Legendre Polynomials are orthogonal to each other:

In[ ]:=
```
TableForm[Table[Integrate[LePo9[[j]] LePo9[[k]], {x, -1, 1}], {j, 1, 10}, {k, 1, 10}]]
```

Out[ ]//TableForm=

2	0	0	0	0	0	0	0	0	0
0	$\frac{2}{3}$	0	0	0	0	0	0	0	0
0	0	$\frac{2}{5}$	0	0	0	0	0	0	0
0	0	0	$\frac{2}{7}$	0	0	0	0	0	0
0	0	0	0	$\frac{2}{9}$	0	0	0	0	0
0	0	0	0	0	$\frac{2}{11}$	0	0	0	0
0	0	0	0	0	0	$\frac{2}{13}$	0	0	0
0	0	0	0	0	0	0	$\frac{2}{15}$	0	0
0	0	0	0	0	0	0	0	$\frac{2}{17}$	0
0	0	0	0	0	0	0	0	0	$\frac{2}{19}$

These polynomials are called Legendre Polynomials. Just to verify that these polynomials have value 1 at x=1

In[ ]:=
```
LePo9 /. {x → 1}
```

Out[ ]=
```
{1, 1, 1, 1, 1, 1, 1, 1, 1, 1}
```

Next I will give a recursive definition of orthogonal polynomials as we do it on the website, using the Gram-Schmidt algorithm:

In[3]:=
```
Clear[GrScOr];
GrScOr[0, x_] = 1; GrScOr[1, x_] = x;
GrScOr[n_, x_] := GrScOr[n, x] =
  Expand[x^n - Sum[Integrate[x^n * GrScOr[k, x], {x, -1, 1}] / Integrate[GrScOr[k, x]^2, {x, -1, 1}] GrScOr[k, x], {k, 0, n - 1}]]
```

In[ ]:=
```
GrScOr[7, x]
```

Out[ ]=

$$-\frac{35x}{429}+\frac{105x^3}{143}-\frac{21x^5}{13}+x^7$$

In[∘]:= (GrScOr[7, t]) /. {t → 1}

Out[∘]= $\frac{16}{429}$

Here is the list of the first 10 of orthogonal polynomials obtained in this way:

In[∘]:= TableForm[Expand[FullSimplify[Table[GrScOr[k, x], {k, 0, 9}]]]]

Out[∘]//TableForm=

1

x

$-\frac{1}{3}+x^2$

$-\frac{3x}{5}+x^3$

$\frac{3}{35}-\frac{6x^2}{7}+x^4$

$\frac{5x}{21}-\frac{10x^3}{9}+x^5$

$-\frac{5}{231}+\frac{5x^2}{11}-\frac{15x^4}{11}+x^6$

$-\frac{35x}{429}+\frac{105x^3}{143}-\frac{21x^5}{13}+x^7$

$\frac{7}{1287}-\frac{28x^2}{143}+\frac{14x^4}{13}-\frac{28x^6}{15}+x^8$

$\frac{63x}{2431}-\frac{84x^3}{221}+\frac{126x^5}{85}-\frac{36x^7}{17}+x^9$

We can use this definition to get an alternative formula for Legendre Polynomials:

In[4]:= LePoR[n_, x_] := Expand[$\frac{1}{(\text{GrScOr}[n, t]) /. \{t \to 1\}}$ GrScOr[n, x]]

In[∘]:= LePoR[7, x]

Out[∘]= $-\frac{35x}{16}+\frac{315x^3}{16}-\frac{693x^5}{16}+\frac{429x^7}{16}$

In[∘]:= LePoR[9, x]

Out[∘]= $\frac{315x}{128}-\frac{1155x^3}{32}+\frac{9009x^5}{64}-\frac{6435x^7}{32}+\frac{12155x^9}{128}$

Here is the list of the first 10 Legendre Polynomials

In[∘]:=
```
TableForm[Expand[FullSimplify[Table[LePoR[k, x], {k, 0, 9}]]]]
```

Out[∘]//TableForm=

$$1$$
$$x$$
$$-\frac{1}{2}+\frac{3x^2}{2}$$
$$-\frac{3x}{2}+\frac{5x^3}{2}$$
$$\frac{3}{8}-\frac{15x^2}{4}+\frac{35x^4}{8}$$
$$\frac{15x}{8}-\frac{35x^3}{4}+\frac{63x^5}{8}$$
$$-\frac{5}{16}+\frac{105x^2}{16}-\frac{315x^4}{16}+\frac{231x^6}{16}$$
$$-\frac{35x}{16}+\frac{315x^3}{16}-\frac{693x^5}{16}+\frac{429x^7}{16}$$
$$\frac{35}{128}-\frac{315x^2}{32}+\frac{3465x^4}{64}-\frac{3003x^6}{32}+\frac{6435x^8}{128}$$
$$\frac{315x}{128}-\frac{1155x^3}{32}+\frac{9009x^5}{64}-\frac{6435x^7}{32}+\frac{12155x^9}{128}$$

Next is a little nice representation of a Legendre Polynomial

In[∘]:=
```
((Last[LePo9] /.
    {a_ * x^n_. :→ DisplayForm@RowBox[{"(", MakeBoxes@a, ")"}] * HoldForm[x^n]}) /.
  HoldForm[x^1] :→ HoldForm[x])
```

Out[∘]=
$$\left(\frac{315}{128}\right)x+\left(-\frac{1155}{32}\right)x^3+\left(\frac{9009}{64}\right)x^5+\left(-\frac{6435}{32}\right)x^7+\left(\frac{12155}{128}\right)x^9$$

Next, we plot Legendre Polynomials:

In[]:=
```
Plot[Evaluate[LePo9], {x, -1, 1}, PlotStyle → ColorData[97, "ColorList"][[Range[1, 10, 1]]],
  PlotRange → {{-1, 1}, {-1, 1}}, AspectRatio → Automatic, ImageSize → 500]
```

Out[]=



In[∘]:=
```
Plot[Evaluate[LePo9], {x, -1, 1},
 PlotStyle → ({Thickness[0.004], #} & /@ ColorData[97, "ColorList"][[Range[1, 10, 1]]]),
 PlotRange → {{-1, 1}, {-1, 1}}, AspectRatio → Automatic, ImageSize → 500]
```

Out[∘]=



In[∘]:=
```
InputForm[ColorData[97, "ColorList"][[Range[1, 2, 1]]]]
```

Out[∘]//InputForm=

```
{RGBColor[0.368417, 0.506779, 0.709798],
 RGBColor[0.880722, 0.611041, 0.142051]}
```

Next, I display the first five Legendre Polynomials of even degree

In[ ]:=
```
Plot[Evaluate[LePo9[[Range[1, 10, 2]]]], {x, -1, 1},
 PlotStyle → InputForm[ColorData[97, "ColorList"][[Range[1, 10, 2]]]],
 PlotRange → {{-1, 1}, {-0.6, 1}}, AspectRatio → Automatic, ImageSize → 500]
```

Out[ ]=



Odd degree

In[ ]:=
```
Plot[Evaluate[LePo9[[Range[2, 10, 2]]]], {x, -1, 1},
 PlotStyle → ColorData[97, "ColorList"][[Range[2, 10, 2]]],
 PlotRange → {{-1, 1}, {-1, 1}}, AspectRatio → Automatic, ImageSize → 500]
```

Out[ ]=


In[ ]:=
```
allL = Plot[Evaluate[LePo9], {x, -1, 1}, PlotPoints → 201,
  PlotStyle → ({Thickness[0.001], #} & /@ ColorData[97, "ColorList"][[Range[1, 10, 1]]]),
  PlotRange → {{-1, 1}, {-1, 1}}, AspectRatio → Automatic, ImageSize → 500]
```

Out[ ]=



In[ ]:=
```
lp = Table[TraditionalForm[Pₙ[x]], {n, 0, 9}]
```

Out[ ]= $\{P_0(x), P_1(x), P_2(x), P_3(x), P_4(x), P_5(x), P_6(x), P_7(x), P_8(x), P_9(x)\}$

In[∘]:=
```
Manipulate[Show[allL, Plot[Evaluate[LePo9[[n + 1]]], {x, -1, 1}, PlotPoints → 201,
   PlotStyle → {Thickness[0.006], ColorData[97, "ColorList"][[n + 1]]}],
  Graphics[{{RGBColor[1, 1, 0], Polygon[
      {{-0.9, 0.74}, {0.7, 0.74}, {0.7, 0.98}, {-0.9, 0.98}, {-0.9, 0.74}}]},
    Text["Legendre polynomial", {-0.12, 0.775}, {1, 0}, BaseStyle → {FontSize → 12}],
    Text[lp[[n + 1]], {-0.09, 0.775}, {-1, 0}, BaseStyle → {FontSize → 14}],
    Text[PolynomialForm[LegendreP[n, x], TraditionalOrder → False],
     {-0.1, 0.9}, {0, 0}, BaseStyle → {FontSize → 10}]}], ImageSize → 400,
  PlotRange → {{-1.005, 1.005}, {-1, 1}}, AspectRatio → Automatic, Axes → False,
  Frame → True, FrameTicks → {{{-1, 0, 1}, {}}, {{-1, 0, 1}, {}}},
  GridLines → {Range[-1, 1, 0.1], Range[-1, 1, 0.1]}, GridLinesStyle →
   {{GrayLevel[0.8], Dashing[{0, Small}]}, {GrayLevel[0.8], Dashing[{0, Small}]}}
 ],
 {{n, 5}, Range[0, 9], Setter}, ControlPlacement → Top, SaveDefinitions → True]
```

Out[∘]=



Below I explore how to display polynomials in traditional form (closed)

Approximations

Next we calculate the approximations of some functions using the first 10 Legendre Polynomials.

First we calculate the squares of the norms of the first 10 Legendre Polynomials:

In[◦]:=
```
Table[{Integrate[LePo9[[k]]^2, {x, -1, 1}], 2/(2 k - 1)}, {k, 1, 10}]
```

Out[◦]=

$$\left\{\{2,2\},\left\{\frac{2}{3},\frac{2}{3}\right\},\left\{\frac{2}{5},\frac{2}{5}\right\},\left\{\frac{2}{7},\frac{2}{7}\right\},\left\{\frac{2}{9},\frac{2}{9}\right\},\left\{\frac{2}{11},\frac{2}{11}\right\},\left\{\frac{2}{13},\frac{2}{13}\right\},\left\{\frac{2}{15},\frac{2}{15}\right\},\left\{\frac{2}{17},\frac{2}{17}\right\},\left\{\frac{2}{19},\frac{2}{19}\right\}\right\}$$

Next we calculate the best approximations of some familiar functions with the first 10 Legendre Polynomials. In the above Manipulations **n** stands for the degree of the approximating polynomial. The approximation by the polynomial of degree **n** is calculated as the projection of a given function f[x] onto the span of the first **n** Legendre Polynomials. The formula is familiar from our work with the dot product, we just replace the dot product with integrals:

$$\sum_{k=0}^{n} \frac{2k+1}{2} \left(\int_{-1}^{1} f[\xi]\, P_k[\xi]\, d\xi \right) P_k[x] \quad (\text{this is a function of } x,\ \text{call it } \mathrm{Proj}_n[x])$$

By the general Pythagorean Theorem the norm of this projection is

$$\sqrt{\int_{-1}^{1} |\,\mathrm{Proj}_n[x]\,|^2\, dx} = \sqrt{\sum_{k=0}^{n} \frac{2k+1}{2} \left(\int_{-1}^{1} f[\xi]\, P_k[\xi]\, d\xi \right)^2}$$

Here $P_k[x]$ is k-th Legendre Polynomial. In the command below it is **LePo9[[k]]**, a member of the list of the Legendre Polynomials that I created earlier.

The calculation of the error is as follows:

$$\sqrt{\int_{-1}^{1} |\,f[x] - \mathrm{Proj}_n[x]\,|^2\, dx} = \sqrt{\int_{-1}^{1} |\,f[x]\,|^2\, dx - \sum_{k=0}^{n} \frac{2k+1}{2} \left(\int_{-1}^{1} f[\xi]\, P_k[\xi]\, d\xi \right)^2}$$

In[78]:=
```
Clear[App];
DynamicModule[{App, coeff}, App = Expand[Table[(2 k - 1)/2 NIntegrate[Cos[Pi x] * LePo9[[k]],
      {x, -1, 1}, WorkingPrecision → 20, AccuracyGoal → 12] * LePo9[[k]], {k, 1, 10}]];
  coeff = Table[(2 k - 1)/2 NIntegrate[Cos[Pi x] * LePo9[[k]], {x, -1, 1},
       WorkingPrecision → 20, AccuracyGoal → 12]^2, {k, 1, 10}];
  Manipulate[Plot[{Cos[Pi x], Total[App[[Range[n + 1]]]]}, {x, -1, 1}, PlotStyle →
     {{Thickness[0.012], RGBColor[0, 0, 0.5]}, {Thickness[0.004], RGBColor[1, 0, 0]}},
    Epilog → {Text["error ≈", {0.8, 1.2}, {0, 0}], Text[N[√NIntegrate[(Cos[Pi x] - Total[
                App[[Range[n + 1]]]])^2, {x, -1, 1}, WorkingPrecision → 20, AccuracyGoal → 12]],
        {0.8, 1.1}, {0, 0}], Text[Sqrt[N[NIntegrate[(Cos[Pi x])^2, {x, -1, 1},
              WorkingPrecision → 20, AccuracyGoal → 12] - Total[coeff[[Range[n + 1]]]]]],
        {0.8, 1}, {0, 0}]}, PlotRange → {-1.3, 1.3}, ImageSize → 500],
   {{n, 5}, Range[0, 9], Setter}, ControlPlacement → Top, SaveDefinitions → True]]
```

Out[78]=



In[77]:=
```
Clear[App];
DynamicModule[{App, coeff}, App = Expand[Table[(2 k - 1)/2 NIntegrate[Sin[Pi x] * LePo9[[k]],
      {x, -1, 1}, WorkingPrecision → 20, AccuracyGoal → 12] * LePo9[[k]], {k, 1, 10}]];
  coeff = Table[(2 k - 1)/2 NIntegrate[Sin[Pi x] * LePo9[[k]], {x, -1, 1},
       WorkingPrecision → 20, AccuracyGoal → 12]^2, {k, 1, 10}];
  Manipulate[Plot[{Sin[Pi x], Total[App[[Range[n + 1]]]]}, {x, -1, 1}, PlotStyle →
     {{Thickness[0.012], RGBColor[0, 0, 0.5]}, {Thickness[0.004], RGBColor[1, 0, 0]}},
    Epilog → {Text["error ≈", {-0.8, 1.2}, {0, 0}], Text[N[√NIntegrate[(Sin[Pi x] - Total[
              App[[Range[n + 1]]]])^2, {x, -1, 1}, WorkingPrecision → 14, AccuracyGoal → 12]],
       {-0.8, 1.1}, {0, 0}], Text[Sqrt[N[NIntegrate[(Sin[Pi x])^2, {x, -1, 1},
            WorkingPrecision → 20, AccuracyGoal → 12] - Total[coeff[[Range[n + 1]]]]]],
       {-0.8, 1}, {0, 0}]}, PlotRange → {-1.3, 1.3}, ImageSize → 500],
   {{n, 5}, Range[0, 9], Setter}, ControlPlacement → Top, SaveDefinitions → True]]
```

Out[77]=



In[76]:=
```
Clear[App];
DynamicModule[{App, coeff}, App = Expand[Table[(2 k - 1)/2 NIntegrate[Exp[x] * LePo9[[k]],
      {x, -1, 1}, WorkingPrecision → 20, AccuracyGoal → 12] * LePo9[[k]], {k, 1, 10}]];
  coeff = Table[(2 k - 1)/2 NIntegrate[Exp[x] * LePo9[[k]], {x, -1, 1},
      WorkingPrecision → 20, AccuracyGoal → 12]^2, {k, 1, 10}];
  Manipulate[Plot[{Exp[x], Total[App[[Range[n + 1]]]]}, {x, -1, 1}, PlotStyle →
     {{Thickness[0.012], RGBColor[0, 0, 0.5]}, {Thickness[0.004], RGBColor[1, 0, 0]}},
    Epilog → {Text["error ≈", {-0.8, 2.85}, {0, 0}], Text[N[√NIntegrate[(Exp[x] - Total[
               App[[Range[n + 1]]]])^2, {x, -1, 1}, WorkingPrecision → 14, AccuracyGoal → 12]],
       {-0.8, 2.7}, {0, 0}], Text[Sqrt[N[NIntegrate[(Exp[x])^2, {x, -1, 1},
            WorkingPrecision → 20, AccuracyGoal → 12] - Total[coeff[[Range[n + 1]]]]]],
       {-0.8, 2.55}, {0, 0}]}, PlotRange → {-.3, 3.3}, ImageSize → 500],
   {{n, 5}, Range[0, 9], Setter}, ControlPlacement → Top, SaveDefinitions → True]]
```

Out[76]=



In[75]:=
```
Clear[App];
DynamicModule[{App, coeff}, App = Expand[Table[(2 k - 1)/2 NIntegrate[x UnitStep[x] * LePo9[[k]],
      {x, -1, 1}, WorkingPrecision → 20, AccuracyGoal → 12] * LePo9[[k]], {k, 1, 10}]];
  coeff = Table[(2 k - 1)/2 NIntegrate[x UnitStep[x] * LePo9[[k]], {x, -1, 1},
      WorkingPrecision → 20, AccuracyGoal → 12]^2, {k, 1, 10}];
  Manipulate[Plot[{x UnitStep[x], Total[App[[Range[n + 1]]]]}, {x, -1, 1}, PlotStyle →
     {{Thickness[0.01], RGBColor[0, 0, 0.5]}, {Thickness[0.005], RGBColor[1, 0, 0]}},
    Epilog → {Text["error ≈", {-0.8, 1.1}, {0, 0}],
      Text[N[√NIntegrate[(x UnitStep[x] - Total[App[[Range[n + 1]]]])^2, {x, -1, 1},
          WorkingPrecision → 14, AccuracyGoal → 12]], {-0.8, 1.05}, {0, 0}],
      Text[Sqrt[N[NIntegrate[(x UnitStep[x])^2, {x, -1, 1}, WorkingPrecision → 20,
           AccuracyGoal → 12] - Total[coeff[[Range[n + 1]]]]]], {-0.8, 1}, {0, 0}]},
    PlotRange → {-0.2, 1.2}, ImageSize → 500, AspectRatio → Automatic],
   {{n, 5}, Range[0, 9], Setter},
   ControlPlacement → Top,
   SaveDefinitions → True]]
```

Out[75]=



In[74]:=
```
Clear[App];
DynamicModule[{App, coeff}, App = Expand[Table[(2 k - 1)/2 NIntegrate[Sign[x] * LePo9[[k]],
      {x, -1, 1}, WorkingPrecision → 20, AccuracyGoal → 12] * LePo9[[k]], {k, 1, 10}]];
  coeff = Table[(2 k - 1)/2 NIntegrate[Sign[x] * LePo9[[k]], {x, -1, 1},
       WorkingPrecision → 20, AccuracyGoal → 12]^2, {k, 1, 10}];
  Manipulate[Plot[{Sign[x], Total[App[[Range[n + 1]]]]}, {x, -1, 1}, PlotStyle → {{Thickness[
          0.01], RGBColor[0, 0, 0.6]}, {Thickness[0.006], RGBColor[1, 0, 0]}}, Exclusions →
       {0}, Prolog → {{PointSize[0.024], RGBColor[0, 0, 0.5], Point[#] & /@ {{0, 0}}}},
     Epilog → {{PointSize[0.024], RGBColor[0, 0, 0.5], Point[#] & /@ {{0, -1}, {0, 1}}},
        {PointSize[0.015], RGBColor[1, 1, 1], Point[#] & /@ {{0, -1}, {0, 1}}},
        Text["error ≈", {-0.8, 1.2}, {0, 0}], Text[N[√NIntegrate[(Sign[x] - Total[App[[Range[
                    n + 1]]]])^2, {x, -1, 1}, WorkingPrecision → 14, AccuracyGoal → 12]], {-0.8,
          1.1}, {0, 0}], Text[Sqrt[N[NIntegrate[(Sign[x])^2, {x, -1, 1}, WorkingPrecision → 20,
             AccuracyGoal → 12] - Total[coeff[[Range[n + 1]]]]]], {-0.8, 1}, {0, 0}]},
     PlotRange → {{-1.3, 1.3}, {-1.3, 1.3}}, ImageSize → 500, AspectRatio → 1/GoldenRatio],
   {{n, 5}, Range[0, 9], Setter}, ControlPlacement → Top, SaveDefinitions → True]]
```

Out[74]=



In[73]:=
```
Clear[App];
DynamicModule[{App, coeff}, App = Expand[Table[(2 k - 1)/2 NIntegrate[Abs[x] * LePo9[[k]],
      {x, -1, 1}, WorkingPrecision → 20, AccuracyGoal → 12] * LePo9[[k]], {k, 1, 10}]];
 coeff = Table[(2 k - 1)/2 NIntegrate[Abs[x] * LePo9[[k]], {x, -1, 1},
      WorkingPrecision → 20, AccuracyGoal → 12]^2, {k, 1, 10}];
 Manipulate[Plot[{Abs[x], Total[App[[Range[n + 1]]]]}, {x, -1, 1}, PlotStyle →
    {{Thickness[0.011], RGBColor[0, 0, 0.5]}, {Thickness[0.006], RGBColor[1, 0, 0]}},
   Epilog → {Text["error ≈", {-0.6, 1.2}, {0, 0}],
     Text[N[√NIntegrate[(Abs[x] - Total[App[[Range[n + 1]]]])^2, {x, -1, 1},
          WorkingPrecision → 14, AccuracyGoal → 12]], {-0.6, 1.15}, {0, 0}],
     Text[Sqrt[N[NIntegrate[(Abs[x])^2, {x, -1, 1}, WorkingPrecision → 20,
           AccuracyGoal → 12] - Total[coeff[[Range[n + 1]]]]]], {-0.6, 1.1}, {0, 0}]},
   PlotRange → {-0.3, 1.3}, ImageSize → 500, AspectRatio → Automatic],
  {{n, 5}, Range[0, 9], Setter},
  ControlPlacement → Top,
  SaveDefinitions → True]]
```

Out[73]=



In[71]:=
```
Clear[App];
DynamicModule[{App, coeff},
 App = Expand[Table[(2 k - 1)/2 NIntegrate[Clip[x, {-1 / 2, 1 / 2}] * LePo9[[k]], {x, -1, 1},
      WorkingPrecision → 20, AccuracyGoal → 12] * LePo9[[k]], {k, 1, 10}]];
 coeff = Table[(2 k - 1)/2 NIntegrate[Clip[x, {-1 / 2, 1 / 2}] * LePo9[[k]],
      {x, -1, 1}, WorkingPrecision → 20, AccuracyGoal → 12]^2, {k, 1, 10}];
 Manipulate[Plot[{Clip[x, {-1 / 2, 1 / 2}], Total[App[[Range[n + 1]]]]}, {x, -1, 1}, PlotStyle →
    {{Thickness[0.01], RGBColor[0, 0, 0.5]}, {Thickness[0.005], RGBColor[1, 0, 0]}},
   Epilog → {Text["error ≈", {-0.7, 0.7}, {0, 0}],
     Text[N[√NIntegrate[(Clip[x, {-1 / 2, 1 / 2}] - Total[App[[Range[n + 1]]]])^2,
         {x, -1, 1}, WorkingPrecision → 14, AccuracyGoal → 12]], {-0.7, 0.65}, {0, 0}],
     Text[Sqrt[N[NIntegrate[(Clip[x, {-1 / 2, 1 / 2}])^2, {x, -1, 1}, WorkingPrecision → 20,
          AccuracyGoal → 12] - Total[coeff[[Range[n + 1]]]]]], {-0.7, 0.6}, {0, 0}]},
   PlotRange → {-0.75, 0.75}, ImageSize → 500, AspectRatio → Automatic],
  {{n, 5}, Range[0, 9], Setter},
  ControlPlacement → Top,
  SaveDefinitions → True]]
```

Out[71]=



Below are some unrelated explorations (closed cells)